

[Description](#)  
[Options](#)[Quick start](#)  
[Remarks and examples](#)[Menu](#)  
[Also see](#)[Syntax](#)

## Description

`twoway spike` displays numerical ( $y,x$ ) data as spikes. `twoway spike` is useful for drawing spike plots of time-series data or other equally spaced data and is useful as a programming tool. For sparse data, also see [\[G-2\] graph bar](#).

## Quick start

A spike plot displaying a spike between ( $y, x$ ) and 0

```
twoway spike y x
```

Same as above, with horizontal spikes

```
twoway spike y x, horizontal
```

Draw spikes from 200 instead of 0

```
twoway spike y x, base(200)
```

Same as above, but set overall appearance to that used by the *Stata Journal*

```
twoway spike y x, base(200) scheme(sj)
```

## Menu

Graphics > Two-way graph (scatter, line, etc.)

## Syntax

```
twoway spike yvar xvar [if] [in] [, options]
```

<i>options</i>	Description
<code>vertical</code>	vertical spike plot; the default
<code>horizontal</code>	horizontal spike plot
<code>base(#)</code>	value to drop to; default is 0
<i>line_options</i>	change look of spike lines
<i>colorvar_options</i>	change color of spike lines based on values of a variable
<i>axis_choice_options</i>	associate plot with alternative axis
<i>twoway_options</i>	titles, legends, axes, added lines and text, by, regions, name, aspect ratio, etc.

All explicit options are *rightmost*, except `vertical` and `horizontal`, which are *unique*; see [G-4] **Concept: repeated options**.

## Options

`vertical` and `horizontal` specify either a vertical or a horizontal spike plot. `vertical` is the default. If `horizontal` is specified, the values recorded in `yvar` are treated as  $x$  values, and the values recorded in `xvar` are treated as  $y$  values. That is, to make horizontal plots, do not switch the order of the two variables specified.

In the `vertical` case, spikes are drawn at the specified `xvar` values and extend up or down from 0 according to the corresponding `yvar` values. If 0 is not in the range of the  $y$  axis, spikes extend up or down to the  $x$  axis.

In the `horizontal` case, spikes are drawn at the specified `xvar` values and extend left or right from 0 according to the corresponding `yvar` values. If 0 is not in the range of the  $x$  axis, spikes extend left or right to the  $y$  axis.

`base(#)` specifies the value from which the spike should extend. The default is `base(0)`; in the above description of options `vertical` and `horizontal`, this default was assumed.

*line\_options* specify the look of the lines used to draw the spikes, including pattern, width, and color; see [G-3] *line\_options*.

*colorvar\_options* specify that the color of the lines used to draw the spikes be determined by the levels of the numeric variable `colorvar`; see [G-3] *colorvar\_options*.

*axis\_choice\_options* associate the plot with a particular  $y$  or  $x$  axis on the graph; see [G-3] *axis\_choice\_options*.

*twoway\_options* are a set of common options supported by all `twoway` graphs. These options allow you to title graphs, name graphs, control axes and legends, add lines and text, set aspect ratios, create graphs over `by()` groups, and change some advanced settings. See [G-3] *twoway\_options*.

# Remarks and examples

Remarks are presented under the following headings:

- Typical use
- Advanced use
- Cautions

## Typical use


We have daily data recording the values for the S&P 500 in 2001:

```
. use https://www.stata-press.com/data/r19/sp500
(S&P 500)
. list date close change in 1/5
```

	date	close	change
1.	02jan2001	1283.27	.
2.	03jan2001	1347.56	64.29004
3.	04jan2001	1333.34	-14.22009
4.	05jan2001	1298.35	-34.98999
5.	08jan2001	1295.86	-2.48999


The example in [G-2] **graph twoway bar** graphed the first 57 observations of these data by using bars. Here is the same graph presented as spikes:

```
. twoway spike change date in 1/57
```



Spikes are especially useful when there are a lot of data. The graph below shows the data for the entire year:


```
. twoway spike change date
```



## Advanced use


The useful thing about `twoway spike` is that it can be combined with other `twoway` plottypes (see [\[G-2\] graph twoway](#)):

```
. twoway line close date || spike change date
```



We can improve this graph by typing

```
. twoway
    line close date, yaxis(1)
    ||
    spike change date, yaxis(2)
    ||,
    ysca(axis(1) r(700 1400)) ylab(1000(100)1400, axis(1))
    ysca(axis(2) r(-50 300)) ylab(-50 0 50, axis(2))
    ytick(-50(25)50, axis(2) grid)
    legend(off)
    xtitle("Date")
    title("S&P 500")
    subtitle("January to December 2001")
    note("Source: Yahoo!Finance and Commodity Systems, Inc.")
    yline(950, axis(1) lstyle(foreground))
```



Concerning our use of

```
ylines(950, axis(1) lstyle(foreground))
```

see [Advanced use: Overlaying in \[G-2\] graph twoway bar](#).

## Cautions

See [Cautions](#) in [\[G-2\] graph twoway bar](#), which applies equally to `twoway spike`.

## Also see

[\[G-2\] graph twoway bar](#) — Two-way bar plots

[\[G-2\] graph twoway dot](#) — Two-way dot plots

[\[G-2\] graph twoway dropline](#) — Two-way dropped-line plots

[\[G-2\] graph twoway scatter](#) — Two-way scatterplots

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2025 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the FAQ on [citing Stata documentation](#).